# $Id$
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: xduugu <xduugu@gmx.com>

pkgbase=mkvtoolnix
pkgname=('mkvtoolnix-cli' 'mkvtoolnix-gui')
pkgver=15.0.0
pkgrel=2
pkgdesc='Set of tools to create, edit and inspect Matroska files'
arch=('i686' 'x86_64')
url='https://www.bunkus.org/videotools/mkvtoolnix/'
license=('GPL')
makedepends=('boost' 'boost-libs' 'docbook-xsl' 'file' 'flac' 'libebml'
             'libmatroska' 'libogg' 'libvorbis' 'qt5-multimedia' 'ruby' 'zlib')
source=("http://www.bunkus.org/videotools/mkvtoolnix/sources/mkvtoolnix-${pkgver}.tar.xz")
sha256sums=('73dc3b6f0a7147e28c06f44427fa0e824d0a4129e40c942d7642d9f451a51195')

build() {
  cd mkvtoolnix-${pkgver}

  ./configure \
    --prefix='/usr' \
    --disable-qt
  rake apps:mkvinfo $MAKEFLAGS
  mv src/mkvinfo{,-cli}

  ./configure \
    --prefix='/usr' \
    --disable-update-check
  rake $MAKEFLAGS
}

package_mkvtoolnix-cli() {
  depends=('boost-libs' 'file' 'flac' 'gcc-libs' 'glibc' 'libogg' 'zlib'
           'libebml.so' 'libmatroska.so' 'libvorbis.so')

  cd mkvtoolnix-${pkgver}

  rake DESTDIR="${pkgdir}" install
  install -m 755 src/mkvinfo-cli "${pkgdir}"/usr/bin/mkvinfo
  rm -rf "${pkgdir}"/usr/share/{applications,icons,mime,mkvtoolnix}
  find "${pkgdir}" -name mkvtoolnix-gui* -delete
}

package_mkvtoolnix-gui() {
  depends=('desktop-file-utils' 'hicolor-icon-theme' 'mkvtoolnix-cli'
           'qt5-multimedia' 'shared-mime-info' 'xdg-utils')
  replaces=('mkvtoolnix-gtk')

  cd mkvtoolnix-${pkgver}

  rake DESTDIR="${pkgdir}" install
  rm -rf "${pkgdir}"/usr/share/locale
  for t in mkv{extract,info,merge,propedit}; do
    find "${pkgdir}" -name $t* -delete
  done
}

# vim: ts=2 sw=2 et:
